

AD-A162 066


FUNCTION PLOTTING SOFTWARE FOR THE IBM-PC AND
IBM-7371/7372 PLOTTERS(U) NAVAL POSTGRADUATE SCHOOL
MONTEREY CA A F ANDRUS NOV 85 NPS55-85-029


1/1

UNCLASSIFIED

F/G 9/2

NL





MICROCOPY RESOLUTION TEST CHART
NATIONAL BUREAU OF STANDARDS-1963-A

NPS55-85-029

NAVAL POSTGRADUATE SCHOOL

Monterey, California

AD-A162 066



DTIC
ELECTE
DEC 10 1985
S
D

TECHNICAL

FUNCTION PLOTTING SOFTWARE FOR THE IBM-PC
AND IBM-7371/7372 PLOTTERS

ALVIN F. ANDRUS

NOVEMBER 1985

Approved for public release; distribution unlimited.

Prepared for:
Naval Postgraduate School
Monterey, CA 93943-5100


DTIC FILE COPY

NAVAL POSTGRADUATE SCHOOL
Monterey, California

Rear Admiral R. H. Shumaker
Superintendent

D. A. Schrady
Provost


This report was prepared by:




Alvin F. Andrus
Associate Professor
Department of Operations Research

Reviewed by:

Released by:



ALAN R. WASHBURN
Chairman of Operations Research



KNEALE T. MARSHALL
Dean of Information/Policy Sciences

Unclassified

SECURITY CLASSIFICATION OF THIS PAGE

AD-A162 066

REPORT DOCUMENTATION PAGE

1a. REPORT SECURITY CLASSIFICATION Unclassified		1b. RESTRICTIVE MARKINGS	
2a. SECURITY CLASSIFICATION AUTHORITY		3. DISTRIBUTION/AVAILABILITY OF REPORT Approved for public release; distribution is unlimited	
2b. DECLASSIFICATION/DOWNGRADING SCHEDULE		4. PERFORMING ORGANIZATION REPORT NUMBER(S)	
5. MONITORING ORGANIZATION REPORT NUMBER(S)		6a. NAME OF PERFORMING ORGANIZATION Naval Postgraduate School	
6b. OFFICE SYMBOL (if applicable)		7a. NAME OF MONITORING ORGANIZATION	
6c. ADDRESS (City, State, and ZIP Code) Monterey, California 93943-5100		7b. ADDRESS (City, State, and ZIP Code)	
8a. NAME OF FUNDING/SPONSORING ORGANIZATION		8b. OFFICE SYMBOL (if applicable)	
9. PROCUREMENT INSTRUMENT IDENTIFICATION NUMBER		10. SOURCE OF FUNDING NUMBERS	
8c. ADDRESS (City, State, and ZIP Code)		PROGRAM ELEMENT NO.	PROJECT NO.
		TASK NO.	WORK UNIT ACCESSION NO.
11. TITLE (Include Security Classification) FUNCTION PLOTTING SOFTWARE FOR THE IBM-PC AND IBM-7371/7372 PLOTTERS			
12. PERSONAL AUTHOR(S) Andrus, Alvin F.			
13a. TYPE OF REPORT Technical	13b. TIME COVERED FROM TO	14. DATE OF REPORT (Year, Month, Day) 85, November	15. PAGE COUNT 21
16. SUPPLEMENTARY NOTATION			
17. COSATI CODES		18. SUBJECT TERMS (Continue on reverse if necessary and identify by block number)	
FIELD	GROUP	SUB-GROUP	
		plotter, IBM-PC	
19. ABSTRACT (Continue on reverse if necessary and identify by block number) This report contains a program and user instructions for plotting functions and line graphs on the IBM-7371 or 7372 color plotter. Program input and operation are illustrated and explained. Examples that illustrate program capabilities are included. The program is written in IBM BASICA.			
20. DISTRIBUTION/AVAILABILITY OF ABSTRACT <input checked="" type="checkbox"/> UNCLASSIFIED/UNLIMITED <input type="checkbox"/> SAME AS RPT <input type="checkbox"/> DTIC USERS		21. ABSTRACT SECURITY CLASSIFICATION	
22a. NAME OF RESPONSIBLE INDIVIDUAL Alvin F. Andrus		22b. TELEPHONE (Include Area Code) (408)646-2413	22c. OFFICE SYMBOL Code 55As

T A B L E O F C O N T E N T S

1. INTRODUCTION.....	1
2. PROGRAM ACCESS.....	1
3. INPUT FUNCTION FILE.....	2
4. PROGRAM STARTUP.....	2
5. MENUS	3
5.1. Menu 1: Plotter Switch Settings.....	3
5.2. Menu-2: Enter File Names.....	3
5.3. Menu-3: Enter Plotting Parameters.....	4
5.4. Menu-4: Enter Function Number To Plot.....	4
5.5. Menu-5: Enter (X,Y) Points.....	4
5.6. Menu-6: Enter Plot Titles.....	5
5.7. Menu-7: Enter Plot Legends.....	5
5.8. Menu-8: Enter X-Axis Labels.....	5
5.9. Menu-9: Enter Y-Axis Labels.....	5
5.10. Menu-10: Edit/Save Input Data.....	5
5.11. Menu-11: Begin Plot.....	5
5.12. Menu-12: Enter Point Plot Parameters.....	5
5.13. Menu-13: Enter Function Plot Parameters.....	6
5.14. Menu-14: Begin Printing.....	6
5.15. Menu-15: Restart/End Program.....	6
6. EXAMPLES.....	7
6.1. Example 1.....	7
6.2. Example 2.....	7
7. APPLICATION EXTENSIONS.....	8
8. APPENDIX.....	12
8.1. APPENDIX A: IBM-PC BASICA Program.....	12
9. DISTRIBUTION.....	20

L I S T O F F I G U R E S

1: Plotter Output for Example 1.....	9
2: Plotter Output for $Y=\tan(X)$ with Jump Function = Y.....	10
3: Plotter Output for $Y=\tan(X)$ with Jump Function = N.....	10
4: Plotter Output for Example 2.....	11

L I S T O F T A B L E S

1: Input Data For Example 1.....	9
2: Input Data For Example 2.....	11

1. INTRODUCTION

Available IBM-PC software for the IBM-7231 and 7232 color plotters has one major deficiency: the software will not accept function definitions as standard input. Plotting functions requires that the functions are input as discrete (X,Y) coordinates.

The program contained and illustrated in this report eliminates the requirement for discrete coordinate inputs in order to plot functions. Continuous single valued functions are defined using the construction rules for standard BASIC expressions. These function definitions then become the program inputs. The program will also accept (X,Y) points and plot a line graph connecting the points. The program is written in IBM BASIC and is included as Appendix A. The program input requirements are:

- a.) Input function file.
- b.) Plotting parameters.
- c.) Line graph points.

The input function file is created as a separate file and contains the function definitions. The plotting parameters and the line graph are provided in response to program interactive menus.

In general the program provides the user with the capability to:

- a.) Plot any selection of five user defined functions.
- b.) Specify line formats, pen selection, grid spacing, labels, legends and plot size.
- c.) Plot a line graph connecting up to 100 (X,Y) points.
- d.) Save the plotting parameters and line graph points in an input data file.

The plot is horizontally centered on 8 1/2 x 11 inch paper. The plotting area is a square with user defined dimensions from three to five inches.

2. PROGRAM ACCESS

The IBM-PC BASIC program is available on diskette for copying from the Operations Research Department Office, Root 272. The diskette contains the program in file IBMPL0T.BAS, sample input function files IBMFEX1.BAS and IBMFEX2.BAS, and sample input data files IBMDEX1 and IBMDEX2 corresponding to Examples 1 and 2 of this report.

Accession For	
NTIS CRA&I	<input checked="" type="checkbox"/>
DTIC TAB	<input type="checkbox"/>
Unannounced	<input type="checkbox"/>
Justification	
By	
Distribution	
Availability Codes	
Dist	Avail and/or Special
A-1	

3. INPUT FUNCTION FILE

In order to plot functions the program requires that the function definitions are constructed as BASIC statements and are contained in an input function file. The format for this file is:

```
5010 Function Definition 1
5020 Function Definition 2
5030 Function Definition 3
5040 Function Definition 4
5050 function Definition 5
```

Each statement begins on a new line. The numbers 6010-6050 are BASIC statement numbers and must occupy columns 1-4. The function definitions are valid BASIC expressions in the form $Y=F(X)$. Each statement can contain 256 characters. The maximum number of function definitions in one input function file is five. When the number of function definitions is less than five, the function definition statements must begin with 5010 and continue in sequence 5020, 5030, etc.

An example of a valid input function file is:

```
5010 Y=SIN(X)
5020 Y=COS(X)
5030 Y=TAN(X)
```

The input function file must be an ASCII file. This file can be created with a word processor, with EDLIN or with BASIC and saved as an ASCII file. This file can have any valid DOS file name but must have ".BAS" as the file name extension.

4. PROGRAM STARTUP

Executing the program requires that:

- a.) The IBM-PC and the IBM Plotter are connected via the COM1 serial port.
- b.) An input function file, if required, is available.

A recommended program startup procedure on the IBM-PC is:

- a.) Place the IBM DOS Diskette in Drive A.
- b.) Turn the IBM-PC on.
- c.) Respond to the Date and Time requests, if required.
- d.) At the A> prompt type BASICA (Return).
- e.) Place the IBM PLOT.BAS diskette in Drive A.
- f.) Type RUN "IBM PLOT" (Return).

From this point on the user is directed by interactive program menus.

5. MENUS

There are fifteen menus that appear in sequence during program execution. These menus prompt the user for inputs or provide information on the status of the program. General information for the input menus is:

- a.) All inputs in a menu are listed with current values.
- b.) As each input is considered for changing it is highlighted with an "*".
- c.) A Y/N in the input prompt indicates the input response is "Y" for Yes and "N" for No.
- d.) The current input value may be retained by responding with a carriage return.
- e.) The input value may be changed by typing a new value followed by a carriage return.
- f.) Corrections to inputs during typing, prior to a carriage return, are made with the delete and backspace keys.
- g.) Key F9 followed by a carriage return will restart the input menu so that input errors, once accepted, can be changed.
- h.) When specified in the menu, Key F10 followed by a carriage return provides a quick advance to the next menu. All inputs that were not changed retain the current displayed values.
- i.) The input processor is not idiot-proof. It works flawlessly when used appropriately. Keep the cursor on the input line.

Each of the menus and associated inputs are explained in the following sections.

5.1. Menu 1: Plotter Switch Settings

Menu 1 provides the switch settings for the IBM-7371 and 7372 plotters. If the switches are not set correctly the program will not run. The plotter should be turned off while the switches are set and then turned on.

5.2. Menu-2: Enter File Names

Menu-2 prompts the user to indicate if an input function file and/or an input data file are to be read. Valid inputs are "Y" and "N". The "Y" response prompts the user for a file name. When these files are read the function definitions and/or the input data are inserted into the program. If functions are to be plotted the function definitions have to read from an input function file which should have been created according to the instructions in Section 3. The input data file is created by the program, if desired, at the end of the input phase.

5.3. Menu-3: Enter Plotting Parameters

Menu-3 prompts the user for plotting parameter inputs. The plotting parameters are described as follows:

XMIN: X-axis minimum value (must be an integer).
XMAX: X-axis maximum value (must be an integer).
YMIN: Y-axis minimum value (must be an integer).
YMAX: Y-axis maximum value (must be an integer).
XSCALE: X-axis division scale factor used to convert X-axis values to noninteger values. For example: If the X-axis is to have the range 0 to .9 then the X-axis inputs are 0 and 9 and the XSCALE input is 10.
YSCALE: Y-axis division scale factor used to convert Y-axis values to noninteger values. (Similar to XSCALE)
SIZE: Size of the square plotting area in inches. Valid inputs are 3 to 5.
NUMBER OF X-AXIS GRIDS: The number of grid space divisions on the X-axis. Grids are drawn. Labels are printed at each grid line. Valid inputs are 1 to 10.
NUMBER OF Y-AXIS GRIDS: The number of grid space divisions on the Y-axis. Grids are drawn. Labels are printed at each grid line. Valid inputs are 1 to 10.
NUMBER OF FUNCTIONS: The number of function definitions in the input function file. Valid inputs are 0 to 5.
JUMP FUNCTION: Determines if the plotting pen is down or up as the pen moves across the plotting area from above to below, or from below to above, the plotting area. Valid inputs are "Y" and "N". "Y" indicates the pen is up. "N" indicates the pen is down. See Example 1.
NUMBER OF (X,Y): The number of individual points to plot in addition to the function definitions. Valid inputs are 0 to 100.
CONNECT POINTS: Determines if the points are connected with a line graph. Valid inputs are "Y" and "N".
MARK POINTS: Determines if points are marked with "*" when plotted. Valid inputs are "Y" and "N".

5.4. Menu-4: Enter Function Number To Plot

Menu-4 prompts the user to indicate for each function if the function is to be plotted. Any combination of the functions can be plotted. Valid inputs are "Y" or "N".

5.5. Menu-5: Enter (X,Y) Points

Menu-5 prompts the user to input the (X,Y) points for plotting. The points are shown on the screen in sets of 15 points. Valid inputs are numbers in the standard integer or decimal format. Points outside of the plotting area will not be plotted.

5.6. Menu-6: Enter Plot Titles.

Menu-6 prompts the user for titles to be printed on the plot. The X-axis title is printed under the X-axis. The Y-axis title is printed along the Y-axis. The Top title is printed above the plot. All titles are centered. Valid inputs are any character except ",", ".".

5.7. Menu-7: Enter Plot Legends

Menu-7 prompts the user to input legend titles for each function to be plotted. The legends are printed to the right of the plot. Valid inputs are any character except ",", ".". The maximum number of characters is 12.

5.8. Menu-8: Enter X-Axis Labels

Menu-8 prompts the user to enter X-axis labels. The labels are centered on the grid lines. Valid inputs are any character except ",", ".".

5.9. Menu-9: Enter Y-Axis Labels

Menu-9 prompts the user to enter Y-axis labels. The labels are centered on the grid lines. Valid inputs are any character except ",", ".".

5.10. Menu-10: Edit/Save Input Data

Menu-10 prompts the user to indicate if the input data is to be edited and/or saved. Valid inputs are "Y" and "N". This is the end of the input phase. If the input data is to be edited for corrections the program returns to Menu-2. If the data is to be saved in a file on a diskette then the user is prompted for a file name.

5.11. Menu-11: Begin Plot

Menu-11 informs the user that the plotter is drawing the plotting area with the specified X-axis and Y-axis grids. No input is required.

5.12. Menu-12: Enter Point Plot Parameters

Menu-12 prompts the user for point plotting inputs for pen selection and line type. Valid inputs for pen selection depend on the number of pen positions on the plotter which are 1 to 6 for the IBM-7372 plotter and 1 to 2 for the IBM-7371 plotter. Valid inputs for line type are included in the menu. Responding with the carriage return provides a solid line. All other responses are either dots or dashes. The plotter will stop and wait for these inputs.

5.13. Menu-13: Enter Function Plot Parameters

Menu-13 prompts the user for function plotting inputs for pen selection and line type. Valid inputs for pen selection depend on the number of pen positions on the plotter which are 1 to 6 for the IBM-7372 plotter and 1 to 2 for the IBM-7371 plotter. Valid inputs for line type are included in the menu. Responding with the carriage return provides a solid line. All other responses are either dots or dashes. The plotter will stop and wait for these inputs prior to plotting each function.

5.14. Menu-14: Begin Printing

Menu-14 informs the user that function and point plotting is complete and that titles and labels are now being printed.

5.15. Menu-15: Restart/End Program

Menu-15 prompts the user to either restart or end the program. Valid inputs are "Y" or "N". If the program is restarted it returns to Menu-2 and all inputs and functions remain in effect until changed. If the program is not restarted it terminates and computer control is returned to DOS.

6. EXAMPLES

Two examples are provided to illustrate program application.

6.1. Example 1

This example plots the trigonometric functions

$Y = \sin(X)$
 $Y = \cos(X)$
 $Y = \tan(X)$

No points are plotted. The input function file and input data file for this example are on the IBMPL0T.BAS diskette in files IBMFEX1.BAS and IBMDEX1. A review of the data is presented in Table 1 and the plotter output is contained in Figure 1.

To specifically illustrate the jump function input this example was run to plot only the function $Y = \tan(X)$ with the jump function input set to "Y" and then set to "N". The plotter output is presented in Figures 2 and 3. The difference in the plotter output is the pen-up and pen-down feature as the pen moves across the discontinuities for Y.

6.2. Example 2

This example plots the eight (X,Y) points

$(X,Y) = (2,.94), (2.5,.78), (3,.65), (3.5,.55)$
 $(4,.52), (4.5,.49), (5,.44), (5.5,.41)$

and the functions

$Y = A * X + B$	$A = -.19$	$B = 1.35$
$Y = B * A^X$	$A = .76$	$B = 1.65$
$Y = B * X^A$	$A = -.83$	$B = 1.65$
$Y = X / (A + B * X)$	$A = -3.35$	$B = 2.83$

The coefficients in the functions were determined from the eight (X,Y) points by least squares with the constraint that $(X,Y) = (1.83, 1.0)$. The input function file and input data file for this example are on the IBMPL0T.BAS diskette in files IBMFEX2.BAS and IBMDEX2. A review of the data is presented in Table 2 and the plotter output is contained in Figure 4. This example compares graphically the derived functions with the original points and illustrates the utility of plotting a line graph along with the function definitions.

7. APPLICATION EXTENSIONS

The BASIC program is included as Appendix A. The program is straight forward and each program menu section is clearly marked. A user familiar with BASIC can extend this program in many ways. The plotting procedures are the heart of the program and many other applications can be built around them. It should not be hard, for instance, to plot and label sequences of line graphs, plot and label bar graphs, perform specific least square calculations and increase the number of functions.

INPUT FUNCTION FILE NAME: IBMFEX1.BAS	X-AXIS TITLE: X (RADIANS)
INPUT DATA FILE NAME: IBMDEX1	Y-AXIS TITLE: Y=F(X)
XMIN: 0	TOP TITLE: TRIG FUNCTIONS
XMAX: 10	LEGEND-1: SIN (X)
YMIN: - 2	LEGEND-2: COS (X)
YMAX: 2	LEGEND-3: TAN (X)
XSCALE: 1	1 X-AXIS LABEL: 0.0
YSCALE: 1	2 X-AXIS LABEL: 2.0
SIZE: 3.5	3 X-AXIS LABEL: 4.0
NUMBER OF X-AXIS GRIDS: 5	4 X-AXIS LABEL: 6.0
NUMBER OF Y-AXIS GRIDS: 4	5 X-AXIS LABEL: 8.0
JUMP FUNCTION: Y	1 Y-AXIS LABEL: -2.0
NUMBER OF (X,Y): 0	2 Y-AXIS LABEL: -1.0
CONNECT POINTS: N	3 Y-AXIS LABEL: 0.0
MARK POINTS: N	4 Y-AXIS LABEL: 1.0
PLOT FUNCTION 1: Y	5 Y-AXIS LABEL: 2.0
PLOT FUNCTION 2: Y	FUNCTION 1 LINE TYPE: 2
PLOT FUNCTION 3: Y	FUNCTION 2 LINE TYPE: 3
	FUNCTION 3 LINE TYPE: 4

Table 1: Input Data For Example 1

TRIG FUNCTIONS




Figure 1: Plotter Output for Example 1

TRIG FUNCTIONS




Figure 2: Plotter Output for $Y = \tan(X)$ with Jump Function = Y

TRIG FUNCTIONS




Figure 3: Plotter Output for $Y = \tan(X)$ with Jump Function = N

INPUT FUNCTION FILE NAME: IBMFEX2.BAS
 INPUT DATA FILE NAME: IBMDEX2

XMIN: 2
 XMAX: 11
 YMIN: 0
 YMAX: 1
 XSCALE: 2
 YSCALE: 1
 SIZE: 3.5

NUMBER OF X-AXIS GRIDS: 7
 NUMBER OF Y-AXIS GRIDS: 5
 JUMP FUNCTION: Y
 NUMBER OF (X,Y): 8
 CONNECT POINTS: Y
 MARK POINTS: Y
 PLOT FUNCTION 1: Y
 PLOT FUNCTION 2: Y
 PLOT FUNCTION 3: Y
 PLOT FUNCTION 4: Y

FUNCTION 1 LINE TYPE: 2
 FUNCTION 2 LINE TYPE: 3
 FUNCTION 3 LINE TYPE: 4
 FUNCTION 4 LINE TYPE: 5

X-AXIS TITLE: TIME LATE (HOURS)
 Y-AXIS TITLE: P(DETECTION)

TDP TITLE: LEAST SQUARE COMPARISON

LEGEND-1: $Y=A \cdot X+B$
 LEGEND-2: $Y=B \cdot A^X$
 LEGEND-3: $Y=B \cdot X^A$
 LEGEND-4: $Y=X/(A+B \cdot X)$

1 X-AXIS LABEL: 2.0
 2 X-AXIS LABEL: 2.5
 3 X-AXIS LABEL: 3.0
 4 X-AXIS LABEL: 3.5
 5 X-AXIS LABEL: 4.0
 6 X-AXIS LABEL: 4.5
 7 X-AXIS LABEL: 5.0
 8 X-AXIS LABEL: 5.5
 1 Y-AXIS LABEL: 0.0
 2 Y-AXIS LABEL: 0.2
 3 Y-AXIS LABEL: 0.4
 4 Y-AXIS LABEL: 0.6
 5 Y-AXIS LABEL: 0.8
 6 Y-AXIS LABEL: 1.0

(X,Y)=(2,.94),(2.5,.78),(3,.65),(3.5,.55)
 (4,.52),(4.5,.49),(5,.44),(5.5,.41)

Table 2: Input Data For Example 2

LEAST SQUARE COMPARISON




Figure 4: Plotter Output for Example 2

8. APPENDIX

8.1. APPENDIX A: IBM-PC BASICA Program

```

10 KEY OFF: CLS: OPTION BASE 1: DIM XP(100),YP(100)
20 MP$="N": LP$="N": JUMP$="Y"
30 XSCALE=1: YSCALE=1: XMIN=0: XMAX=1: YMIN=0: YMAX=1: SIZE=4: XG=1: YG=1
40 FOR I=1 TO 10: KEY I, "": NEXT I
50 '-----MENU-1
60 GOSUB 3300
70 LOCATE 3,24: PRINT "MENU-1: PLOTTER SWITCH SETTINGS"
80 LOCATE 5,25: PRINT "PROFESSOR A. ANDRUS, CODE 55AS"
90 LOCATE 6,27: PRINT "NAVAL POSTGRADUATE SCHOOL"
100 LOCATE 8,15: PRINT "TURN PLOTTER OFF. SET SWITCHES AS FOLLOWS: (I=ON, O=OFF)"
110 LOCATE 10,15: PRINT "HP-7470-A and IBM-7371"
120 LOCATE 12,15: PRINT "SWITCH: S2 S1 Y US B4 B3 B2 B1"
130 LOCATE 13,15: PRINT "SET: I I O I I O I O"
140 LOCATE 15,15: PRINT "HP-7475-A and IBM-7372"
150 LOCATE 17,15: PRINT "SWITCH: S2 S1 Y US A3/A4 B4 B3 B2 B1"
160 LOCATE 18,15: PRINT "SET: I I O I A4 I O I O"
170 LOCATE 20,15: PRINT "TURN PLOTTER ON."
180 LOCATE 23,26: PRINT ".. PRESS ";CHR$(17);CHR$(196);CHR$(217);" TO CONTINUE ..";
190 A$=INKEY$: IF A$<>CHR$(13) THEN 190 ELSE CLS
200 OPEN "COM1: 9600,S,7,1,RS,CS65535,DS,CD" AS #1
210 PRINT #1,"IN;SP 1;PA 500,500;"
220 '-----MENU-2
230 GOSUB 3300
240 LOCATE 3,27: PRINT "MENU-2: ENTER FILE NAMES"
250 LOCATE 10,20: INPUT "* INPUT FUNCTION FILE NAME Y/N ";A$
260 IF A$="Y" OR A$="y" THEN 270 ELSE 290
270 LOCATE 11,20: INPUT " FILE NAME = ",FF$
280 IF FF$<>" " THEN CHAIN MERGE FF$, 290,ALL
290 LOCATE 13,20: INPUT "* INPUT DATA FILE NAME Y/N ";A$
300 IF A$="Y" OR A$="y" THEN 310 ELSE 460
310 LOCATE 14,20: INPUT " FILE NAME = ",DF$
320 IF DF$="" THEN 460 ELSE OPEN DF$ FOR INPUT AS 2
330 INPUT#2,XMIN,XMAX,YMIN,YMAX,XSCALE,YSCALE,SIZE,XG,YG,IP,NF,NXL,NYL
340 FOR I=1 TO IP: INPUT#2,XP(I),YP(I): NEXT I
350 INPUT#2,LP$: INPUT#2,MP$
360 INPUT#2,XLABEL$: INPUT#2,YLABEL$
370 INPUT#2,TOP$: INPUT#2,JUMP$
380 FOR I=1 TO NF: INPUT#2,KF$(I): NEXT I
390 FOR I=1 TO NXL: INPUT#2,XL$(I): NEXT I
400 FOR I=1 TO NYL: INPUT#2,YL$(I): NEXT I
410 FOR I=1 TO NF: INPUT#2,L$(I): NEXT I
420 CLOSE#2: GOTO 460
430 '-----MENU-3
440 KEY(10) OFF: RETURN 960: 'F-TRAP SUBROUTINE
450 KEY(9) OFF: KEY(10) OFF: RETURN 460
460 KEY(9) ON: ON KEY(9) GOSUB 450: KEY(10) ON: ON KEY(10) GOSUB 440
470 GOSUB 3300: GOSUB 3600: GOSUB 3560
480 LOCATE 3,25: PRINT "MENU-3: ENTER PLOT PARAMETERS"
490 LOCATE 7,15: PRINT "XMIN =";XMIN
500 LOCATE 8,15: PRINT "XMAX =";XMAX

```

```

510 LOCATE 9,15: PRINT "YMIN =";YMIN
520 LOCATE 10,15: PRINT "YMAX =";YMAX
530 LOCATE 11,15: PRINT "XSCALE=";XSCALE
540 LOCATE 12,15: PRINT "YSCALE=";YSCALE
550 LOCATE 13,15: PRINT "SIZE =";SIZE
560 LOCATE 14,15: PRINT "NUMBER OF X-AXIS GRIDS =";XG
570 LOCATE 15,15: PRINT "NUMBER OF Y-AXIS GRIDS =";YG
580 LOCATE 16,15: PRINT "NUMBER OF FUNCTIONS =";NF
590 LOCATE 17,15: PRINT "JUMP FUNCTION Y/N = ";JUMP$
600 LOCATE 18,15: PRINT "NUMBER OF (X,Y) TO PLOT=";IP
610 LOCATE 19,15: PRINT "CONNECT POINTS Y/N = ";LP$
620 LOCATE 20,15: PRINT "MARK POINTS Y/N = ";MP$
630 LOCATE 7,13: PRINT "*"
640 LOCATE 7,50: INPUT XMIN$: IF XMIN$<>" THEN XMIN=VAL(XMIN$)
650 LOCATE 8,13: PRINT "*"
660 LOCATE 8,50: INPUT XMAX$: IF XMAX$<>" THEN XMAX=VAL(XMAX$)
670 LOCATE 9,13: PRINT "*"
680 LOCATE 9,50: INPUT YMIN$: IF YMIN$<>" THEN YMIN=VAL(YMIN$)
690 LOCATE 10,13: PRINT "*"
700 LOCATE 10,50: INPUT YMAX$: IF YMAX$<>" THEN YMAX=VAL(YMAX$)
710 LOCATE 11,13: PRINT "*"
720 LOCATE 11,50: INPUT XSCALE$: IF XSCALE$<>" THEN XSCALE=VAL(XSCALE$)
730 IF XSCALE<=0 THEN XSCALE=1
740 LOCATE 12,13: PRINT "*"
750 LOCATE 12,50: INPUT YSCALE$: IF YSCALE$<>" THEN YSCALE=VAL(YSCALE$)
760 IF YSCALE<=0 THEN YSCALE=1
770 LOCATE 13,13: PRINT "*"
780 LOCATE 13,50: INPUT SIZE$
790 LOCATE 14,13: PRINT "*"
800 LOCATE 14,50: INPUT XG$: IF XG$<>" THEN XG=VAL(XG$)
810 IF XG<=0 THEN XG=1 ELSE IF XG>=10 THEN XG=10
820 LOCATE 15,13: PRINT "*"
830 LOCATE 15,50: INPUT YG$: IF YG$<>" THEN YG=VAL(YG$)
840 IF YG<=0 THEN YG=1 ELSE IF YG>=10 THEN YG=10
850 LOCATE 16,13: PRINT "*"
860 LOCATE 16,50: INPUT NF$: IF NF$<>" THEN NF=VAL(NF$)
870 LOCATE 17,13: PRINT "*"
880 LOCATE 17,50: INPUT T$: IF T$<>" THEN JUMP$=T$
890 LOCATE 18,13: PRINT "*"
900 LOCATE 18,50: INPUT IP$: IF IP$<>" THEN IP=VAL(IP$)
910 IF IP>100 THEN IP=100
920 LOCATE 19,13: PRINT "*"
930 LOCATE 19,50: INPUT A$: IF A$<>" THEN LP$=A$
940 LOCATE 20,13: PRINT "*"
950 LOCATE 20,50: INPUT A$: IF A$<>" THEN MP$=A$
960 IF SIZE$<>" THEN SIZE=VAL(SIZE$)
970 IF SIZE<3 THEN SIZE=3 ELSE IF SIZE>5 THEN SIZE=5
980 X0=500*(10.3-SIZE)-157: Y0=500*(7.65-SIZE)+188: Z=SIZE*1000

```

```

990 -----MENU-4
1000 IF NF=0 THEN 1140 ELSE 1030
1010 KEY(10) OFF: RETURN 1140: 'F-TRAP SUBROUTINE
1020 KEY(9) OFF: KEY(10) OFF: RETURN 1030
1030 KEY(9) ON: ON KEY(9) GOSUB 1020: KEY(10) ON: ON KEY(10) GOSUB 1010
1040 GOSUB 3300: GOSUB 3600: GOSUB 3560
1050 LOCATE 3,21: PRINT "MENU-4: ENTER FUNCTION NUMBER TO PLOT"
1060 FOR I=1 TO NF
1070   LOCATE 10+I,20: PRINT "PLOT FUNCTION";I;" Y/N : ";KF$(I)
1080 NEXT I
1090 FOR I=1 TO NF
1100   LOCATE 10+I,18: PRINT "*"
1110   LOCATE 10+I,50: INPUT T$: IF T$(">") THEN KF$(I)=T$
1120 NEXT I
1130 -----MENU-5
1140 IF IP=0 THEN 1450 ELSE 1170
1150 KEY(10) OFF: RETURN 1450: 'F-TRAP SUBROUTINE
1160 KEY(9) OFF: KEY(10) OFF: RETURN 1170
1170 KEY(9) ON: ON KEY(9) GOSUB 1160: KEY(10) ON: ON KEY(10) GOSUB 1150
1180 GOSUB 3300: GOSUB 3600: GOSUB 3560
1190 FOR J=1 TO IP STEP 15
1200   LOCATE 3,27: PRINT "MENU-5: ENTER (X,Y) POINTS"
1210   IF J+14>IP THEN JE=IP ELSE JE=J+14
1220   K=0
1230   FOR I=J TO JE
1240     K=K+1
1250     LOCATE 5+K,3: PRINT I
1260     LOCATE 5+K,7: PRINT "X=";XP(I)
1270     LOCATE 5+K,25: PRINT "Y=";YP(I)
1280   NEXT I
1290   K=0
1300   FOR I=J TO JE
1310     K=K+1
1320     LOCATE 5+K,3: PRINT "*"
1330     LOCATE 5+K,43: INPUT "X=";XP$
1340     IF XP$="P" OR XP$="p" THEN 1400
1350     LOCATE 5+K,61: INPUT "Y=";YP$
1360     IF YP$="P" OR YP$="p" THEN 1400
1370     IF XP$(">") THEN XP(I)=VAL(XP$)
1380     IF YP$(">") THEN YP(I)=VAL(YP$)
1390   NEXT I
1400 NEXT J
1410 GOTO 1450

```

```

1420 '-----MENU-6
1430 KEY(10) OFF: RETURN 1580: 'F-TRAP SUBROUTINE
1440 KEY(9) OFF: KEY(10) OFF: RETURN 1450
1450 KEY(9) ON: ON KEY(9) GOSUB 1440: KEY(10) ON: ON KEY(10) GOSUB 1430
1460 GOSUB 3300: GOSUB 3600: GOSUB 3560
1470 LOCATE 3,27: PRINT "MENU-6: ENTER PLOT TITLES"
1480 LOCATE 10,20: PRINT "X-AXIS TITLE      = ";XLABEL$
1490 LOCATE 11,20: PRINT "Y-AXIS TITLE      = ";YLABEL$
1500 LOCATE 12,20: PRINT "TOP OF GRAPH TITLE = ";TOP$
1510 LOCATE 14,18: INPUT "* NEW X-AXIS TITLE  = ",T$
1520 IF T$<>" THEN XLABEL$=T$
1530 LOCATE 15,18: INPUT "* NEW Y-AXIS TITLE  = ",T$
1540 IF T$<>" THEN YLABEL$=T$
1550 LOCATE 16,18: INPUT "* NEW TOP TITLE    = ",T$
1560 IF T$<>" THEN TOP$=T$
1570 '-----MENU-7
1580 IF NF=0 THEN 1760 ELSE 1610
1590 KEY(10) OFF: RETURN 1760: 'F-TRAP SUBROUTINE
1600 KEY(9) OFF: KEY(10) OFF: RETURN 1610
1610 KEY (9) ON: ON KEY(9) GOSUB 1600: KEY(10) ON: ON KEY(10) GOSUB 1590
1620 GOSUB 3300: GOSUB 3600: GOSUB 3560
1630 LOCATE 3,27: PRINT "MENU-7: ENTER PLOT LEGENDS"
1640 K=7
1650 FOR I=1 TO NF
1660 K=K+1
1670 LOCATE K,15: PRINT "LEGEND-";I;" = ";L$(I)
1680 NEXT I
1690 K=7
1700 FOR I=1 TO NF
1710 K=K+1
1720 LOCATE K,13: PRINT "*"
1730 LOCATE K,45: INPUT "NEW LEGEND = ",T$: IF T$<>" THEN L$(I)=T$
1740 NEXT I
1750 '-----MENU-8
1760 KEY(9) ON: ON KEY(9) GOSUB 1790: KEY(10) ON: ON KEY(10) GOSUB 1780
1770 GOSUB 3300: GOSUB 3600: GOSUB 3560: GOTO 1800
1780 KEY(10) OFF: RETURN 1870: 'F-TRAP SUBROUTINE
1790 KEY(9) OFF: KEY (10) OFF: RETURN 1760
1800 LOCATE 3,26: PRINT "MENU-8: ENTER X-AXIS LABELS"
1810 IF X6=0 THEN NXL=2 ELSE NXL=X6+1
1820 FOR I=1 TO NXL: LOCATE 8+I,10: PRINT I;"X-AXIS LABEL =";XL$(I): NEXT I
1830 FOR I=1 TO NXL
1840 LOCATE 8+I,8: PRINT "*"
1850 LOCATE 8+I,45: INPUT "NEW LABEL =",T$: IF T$<>" THEN XL$(I)=T$
1860 NEXT I
1870 FOR I=1 TO NXL
1880 XL(I)=LEN(XL$(I)): IF XL(I)>XLMAX THEN XLMAX=XL(I)
1890 NEXT I

```

```

1900 '-----MENU-9
1910 KEY(9) ON: ON KEY(9) GOSUB 1940: KEY(10) ON: ON KEY(10) GOSUB 1930
1920 GOSUB 3300: GOSUB 3600: GOSUB 3560: GOTO 1950
1930 KEY(10) OFF: RETURN 2020: 'F-TRAP SUBROUTINE
1940 KEY(9) OFF: KEY(10) OFF: RETURN 1910
1950 LOCATE 3,26: PRINT "MENU-9: ENTER Y-AXIS LABELS"
1960 IF Y6=0 THEN NYL=2 ELSE NYL=Y6+1
1970 FOR I=1 TO NYL: LOCATE 8+I,10: PRINT I;"Y-AXIS LABEL =";YL$(I): NEXT I
1980 FOR I=1 TO NYL
1990 LOCATE 8+I,8: PRINT "*"
2000 LOCATE 8+I,45: INPUT "NEW LABEL =",T$: IF T$(>)" THEN YL$(I)=T$
2010 NEXT I
2020 FOR I=1 TO NYL
2030 YL(I)=LEN(YL$(I)): IF YL(I)>YLMAX THEN YLMAX=YL(I)
2040 NEXT I
2050 '-----MENU-10
2060 KEY(9) ON: ON KEY(9) GOSUB 2070: KEY(10) OFF: GOSUB 3300: GOTO 2080
2070 KEY(9) OFF: RETURN 2060: 'F-TRAP SUBROUTINE
2080 LOCATE 3,25: PRINT "MENU-10: EDIT/SAVE INPUT DATA"
2090 LOCATE 12,20: INPUT "* EDIT INPUT DATA Y/N ";A$
2100 IF A$="Y" OR A$="y" THEN 230
2110 LOCATE 14,20: INPUT "* SAVE INPUT DATA Y/N ";A$
2120 IF A$="Y" OR A$="y" THEN 2130 ELSE 2260
2130 LOCATE 15,20: INPUT " FILE NAME = ",DF$: IF DF$="" THEN 2260
2140 OPEN DF$ FOR OUTPUT AS 2
2150 PRINT#2,XMIN, XMAX,YMIN,YMAX,XSCALE,YSCALE,SIZE,X6,Y6,IP,NF,NXL,NYL
2160 FOR I=1 TO IP: PRINT#2,XP(I),YP(I): NEXT I
2170 PRINT#2,LP$: PRINT#2,MP$
2180 PRINT#2,XLABEL$: PRINT#2,YLABEL$
2190 PRINT#2,TOP$: PRINT#2,JUMP$
2200 FOR I=1 TO NF: PRINT#2,KF$(I): NEXT I
2210 FOR I=1 TO NXL: PRINT#2,XL$(I): NEXT I
2220 FOR I=1 TO NYL: PRINT#2,YL$(I): NEXT I
2230 FOR I=1 TO NF: PRINT#2,L$(I): NEXT I
2240 CLOSE#2
2250 '-----MENU-11
2260 GOSUB 3300
2270 LOCATE 3,30: PRINT "MENU-11: BEGIN PLOT"
2280 LOCATE 12,28: PRINT "PLOTING AXES AND GRIDS"
2290 PRINT#1,"IP",X0,Y0,X0+Z,Y0+Z,";"
2300 PRINT#1,"SC",XMIN,XMAX,YMIN,YMAX,";"
2310 PRINT#1,"PA",XMIN,YMIN,";PD;"
2320 PRINT#1,"PA",XMIN,YMAX,XMAX,YMAX,XMAX,YMIN,XMIN,YMIN,";PU;"
2330 PRINT#1,"IW",X0,Y0,X0+Z,Y0+Z,";"
2340 PRINT#1,"TL 100,0;"
2350 DXG=(XMAX-XMIN)/X6
2360 FOR I=1 TO X6-1: PRINT#1,"PA",I*DXG+XMIN,YMIN,";PD;XT;PU;": NEXT I
2370 DYG=(YMAX-YMIN)/Y6
2380 FOR I=1 TO Y6-1: PRINT #1,"PA",XMIN,I*DYG+YMIN,";PD;YT;PU;": NEXT I

```

```

2390 -----MENU-12
2400 IF IP=0 THEN 2560
2410 GOSUB 3300
2420 LOCATE 3,22: PRINT "MENU-12: ENTER POINT PLOT PARAMETERS"
2430 GOSUB 3390
2440 PRINT#1, "SI .38,.54;"
2450 IF (MP$="Y" OR MP$="y") AND (LP$="Y" OR LP$="y") THEN 2490
2460 IF (MP$="Y" OR MP$="y") AND (LP$="N" OR LP$="n") THEN 2510
2470 IF (MP$="N" OR MP$="n") AND (LP$="Y" OR LP$="y") THEN 2530
2480 GOTO 2560
2490 FOR I=1 TO IP: PRINT#1, "SM;PA",XP(I)*XSCALE,YP(I)*YSCALE,";PD;": NEXT I
2500 PRINT#1, "SM;PU;": GOTO 2560
2510 FOR I=1 TO IP: PRINT#1, "SM;PA",XP(I)*XSCALE,YP(I)*YSCALE,";": NEXT I
2520 PRINT#1, "SM;PU;": GOTO 2560
2530 FOR I=1 TO IP: PRINT#1, "PA",XP(I)*XSCALE,YP(I)*YSCALE,";PD;": NEXT I
2540 PRINT#1, "SM;PU;": GOTO 2560
2550 -----MENU-13
2560 PRINT#1, "SI;": KF=0
2570 FOR INF=1 TO NF
2580   IF KF$(INF)="N" OR KF$(INF)="n" THEN 2690 ELSE KF=KF+1
2590   GOSUB 3300
2600   LOCATE 3,20: PRINT "MENU-13: ENTER FUNCTION PLOT PARAMETERS"
2610   LOCATE 6,27: PRINT "PLOTING FUNCTION ";INF
2620   GOSUB 3390
2630   X=XMIN/XSCALE
2640   ON INF GOSUB 5010,5020,5030,5040,5050
2650   Y=Y*YSCALE: X=X*XSCALE
2660   IF JUMP$="N" OR JUMP$="n" THEN 2680
2670   IF Y<YMIN OR Y>YMAX THEN YC$="YES" ELSE YC$="NO"
2680   PRINT #1,"PA",X,Y,";PD;"
2690   FOR XT=XMIN TO XMAX STEP (XMAX-XMIN)/(2/10)
2700     X=XT/XSCALE
2710     ON INF GOSUB 5010,5020,5030,5040,5050
2720     Y=Y*YSCALE: X=X*XSCALE
2730     IF JUMP$="N" OR JUMP$="n" THEN PRINT#1, "PA": GOTO 2770
2740     IF Y<YMIN OR Y>YMAX THEN YC1$="YES" ELSE YC1$="NO"
2750     IF YC$="YES" AND YC1$="YES" THEN PRINT#1,"PU;PA" ELSE PRINT#1,"PD;PA"
2760     YC$=YC1$
2770     PRINT #1,USING "####.####";X,Y
2780     PRINT #1,";"
2790   NEXT XT
2800   IF L$(INF)=" " THEN 2880
2810   PRINT#1,"IW",X0,Y0,X0+Z+1000,Y0+Z,";"
2820   PRINT#1,"SC;SI .13,.16;PU;PA",X0+Z+50,Y0+Z-300*KF,";"
2830   PRINT#1,"LB";L$(INF),CHR$(3)
2840   PRINT#1,"PA",X0+Z+50,Y0+Z-300*KF-35,";"
2850   PRINT#1,"PD;PA",X0+Z+750,Y0+Z-300*KF-35,";"
2860   PRINT#1,"PU;SC",XMIN,XMAX,YMIN,YMAX,";SI .19,.27;"
2870   PRINT #1,"IW",X0,Y0,X0+Z,Y0+Z,";"
2880   PRINT#1, "PU;"
2890 NEXT INF
2900 PRINT #1,"PU;IP;IW;SC;SP 1;"

```

```

2910 '-----MENU-14
2920 GOSUB 3300
2930 LOCATE 3,28: PRINT "MENU-14: BEGIN PRINTING"
2940 LOCATE 12,27: PRINT "PRINTING LABELS AND TITLES"
2950 LX=LEN(XLABEL$)
2960 NX=(Z*9/1000-LX)/2
2970 PRINT#1, "SI .13,.16;"
2980 FOR I=1 TO NXL
2990 IF XL$(I)="" THEN 3020
3000 PRINT#1,"PA",X0+Z/(NXL-1)*(I-1),Y0-100,";"
3010 PRINT#1,"CP",-XL(I)/2,0,";LB";XL$(I),CHR$(3)
3020 NEXT I
3030 IF XLABEL$="" THEN 3070
3040 PRINT#1, "SI .19,.27;"
3050 IF XLMAX=0 THEN YT=100 ELSE YT=350
3060 PRINT #1,"PA",X0,Y0-YT,";CP",NX,0,";LB";XLABEL$,CHR$(3)
3070 LY=LEN(YLABEL$)
3080 NY=(Z*9/1000-LY)/2
3090 PRINT#1, "SI .13,.16;"
3100 FOR I=1 TO NYL
3110 IF YL$(I)="" THEN 3150
3120 IF I=1 THEN YT=0 ELSE YT=-.2
3130 PRINT #1,"PA",X0-20,Y0+Z/(NYL-1)*(I-1),";"
3140 PRINT #1,"CP",-YL(I),YT,";LB";YL$(I),CHR$(3)
3150 NEXT I
3160 IF YLABEL$="" THEN 3190 ELSE XT=YLMAX*1000/9+100
3170 PRINT#1, "SI .19,.27;"
3180 PRINT #1,"PA",X0-XT,Y0,";DI 0,1;CP",NY,0,";LB";YLABEL$,CHR$(3),"DI 1,0;"
3190 IF TOP$="" THEN 3220 ELSE PRINT#1, "SI .27,.40;"
3200 PRINT#1,"PA",X0,Y0+Z+300,";"
3210 PRINT#1,"CP",(Z*6.2/1000-LEN(TOP$))/2,0,";LB";TOP$,CHR$(3)
3220 PRINT#1, "PU;SC;PA 0,0;"
3230 CLOSE#1
3240 '-----MENU-15
3250 GOSUB 3300
3260 LOCATE 3,26: PRINT "MENU-15: RESTART/END PROGRAM"
3270 LOCATE 12,25: INPUT "* RESTART PROGRAM Y/N ";T$
3280 IF T$="Y" OR T$="y" THEN CLOSE#1: GOTO 200
3290 CLOSE#1: SYSTEM: END
3300 '.....SCREEN BORDER SUBROUTINE
3310 CLS
3320 LOCATE 2,1: PRINT CHR$(201);STRING$(77,205);CHR$(187)
3330 FOR I=3 TO 23
3340 LOCATE I,1: PRINT CHR$(186)
3350 LOCATE I,79: PRINT CHR$(186);
3360 NEXT I
3370 LOCATE 24,1: PRINT CHR$(200);STRING$(77,205);CHR$(188);
3380 RETURN

```



```

3390 .....PEN SELECTION SUBROUTINE
3400 PRINT#1, "SP;"
3410 LOCATE 8,25: INPUT "* SELECT PEN COLOR POSITION: ";PCP
3420 LOCATE 10,25: PRINT " SELECT LINE TYPE: "
3430 LOCATE 12,25: PRINT "C/R: SOLID
3440 LOCATE 13,25: PRINT " 0: DOTS AT PLOTTED POINTS"
3450 LOCATE 14,25: PRINT " 1: DOTS EQUALLY SPACED"
3460 LOCATE 15,25: PRINT " 2: SHORT DASH"
3470 LOCATE 16,25: PRINT " 3: LONG DASH"
3480 LOCATE 17,25: PRINT " 4: LONG DASH AND DOT"
3490 LOCATE 18,25: PRINT " 5: LONG AND SHORT DASH"
3500 LOCATE 19,25: PRINT " 6: LONG AND 2 SHORT DASHES"
3510 LOCATE 10,25: PRINT "*"
3520 LOCATE 10,54: INPUT LT$
3530 PRINT#1, "SP";PCP,";"
3540 IF LT$="" THEN PRINT#1, "LT;" ELSE PRINT#1, "LT",VAL(LT$),";"
3550 RETURN
3560 .....F10 LINE SUBROUTINE
3570 LOCATE 23,25
3580 PRINT "PRESS F10 AND ";CHR$(17);CHR$(196);CHR$(217);" FOR QUICK ADVANCE"
3590 RETURN
3600 .....F9 LINE SUBROUTINE
3610 LOCATE 22,25
3620 PRINT "PRESS F9 AND ";CHR$(17);CHR$(196);CHR$(217);" TO REPEAT MENU"
3630 RETURN
5000 .....FUNCTION SUBROUTINES
5010 RETURN
5015 RETURN
5020 RETURN
5025 RETURN
5030 RETURN
5035 RETURN
5040 RETURN
5045 RETURN
5050 RETURN
5055 RETURN

```

DISTRIBUTION LIST

	<u>No. of Copies</u>
Defense Technical Information Center Cameron Station Alexandria, VA 22314	2
Library (Code 0142) Naval Postgraduate School Monterey, CA 93943-5100	2
Office of Research Administration (Code 012) Naval Postgraduate School Monterey, CA 93943-5100	1
Navy Tactical Support Activity P. O. Box 1042 Silver Spring, MD 20910	2
Naval Air Development Center Attn: Dr. Martin Leonardo Warminster, PA 18974	2
Center for Naval Analyses 2000 Beauregard Street Alexandria, VA 22311	1
Operations Research Center, Room E40-164 Massachusetts Institute of Technology Attn: R. C. Larson and J. F. Shapiro Cambridge, MA 02139	1
Library (Code 55) Naval Postgraduate School Monterey, CA 93943-5100	1
Department of Operations Research Code 55 Naval Postgraduate School Monterey, CA 93943-5100	10
Professor A. Andrus (Code 55As) Naval Postgraduate School Monterey, CA 93943-5100	20

END

FILMED

1-86

DTIC